


★ネットで挑戦★

数字パズルゲームを 作ろう

上田市マルチメディア情報センター／十勇士パソコンクラブ／PCN上田

1



数字を入れかえて、同じ数字をペアにして消す
数字パズルゲームを作ろう

2

IchigoJam web

```
57328281
><
0
```

QRコード



IchigoJam webでも作れるよ!
fukuno.jig.jp/app/IchigoJam/

KEY ESC EXPORT IMPORT FULL I/O AUDIO ON

3


(1) 8個の数字を表示

```
10 CLV:CLS:VIDEO 3:L=7
20 GSB 100
90 END
100 LC 1,0:FOR K=0 TO L
110 ?[K]::NEXT:RTN
```

5行のプログラムを入力します

4

入力できたら、F5で実行



5


(2) 8個の数字をランダムにする

```
15 FOR I=0 TO L:GSB 120:NEXT  
120 [L+1]=1+RND(9)  
130 FOR K=J TO L:[K]=[K+1]  
140 NEXT:RTN
```

4行のプログラムを入力します

6

入力できたら、F5で実行



7

(3) 操作カーソルを表示

スペース


スペース

```
30 LC X,1:"? "><"
40 K=INKEY()
50 X=X-(K=28)*(X>0)+(K=29)*(X<L-1)
90 WAIT 6:GOTO 30
```

4行追加します

8

入力できたら、F5で実行



9


(4)スペースキーで数字を入れかえる

```
60  IF  K=32  GSB  150  
150  A=[X]:[X]=[X+1]:[X+1]=A  
160  GSB  100  
170  RTN
```

4行追加します

10

入力できたら、F5で実行



11


(5) 同じ数字がとなりになったら消す

```
165 GSB 200  
200 P=1:FOR J=0 TO L-1  
210 IF [J]=[J+1] GSB 230:J=-1  
220 NEXT:RTN  
230 LC J+1,0:?"**":BEEP:WAIT 10  
240 RTN
```

6行追加します

12

入力できたら、F5で実行



13

(6) 数字を消して、右から新しく出す

235 GSB 120:GSB 120:GSB 100

1行入力します

14

入力できたら、F5で実行

98365217

> <

同じ数字を2個消すと、
数字が左へ移動して、
新しい数字が右から2個出てくる。
ペアが複数あると連鎖する。

15

(7)最初にペアをチェックする

15 GSB 200

1行追加(書きかえ)します

16

入力できたら、F5で実行

```
17548693  
><
```

ペアになる数字が
最初にチェックされて消される

17

(8)スコアを出す


```
25 S=0:GSB 238  
237 S=S+P:P=P*2  
238 LC 0,3: ?S; " "
```

スペース2文字

3行追加します

18

入力できたら、F5で実行



数字を消すとスコアが加算
連鎖すると1→2→4→8と
点数が倍々になる

19

●数字パズルゲーム 完成(1)

```
10 CLV:CLS:VIDEO 3:L=7 初期設定
15 GSB 200 ペア数字のチェック
20 GSB 100 数字を表示
25 S=0:GSB 238 スコアを表示
30 LC X,1:"?" "><" カーソルを表示
40 K=INKEY() キー入力読み取り カーソルを左右移動
50 X=X-(K=28)*(X>0)+(K=29)*(X<L-1)
60 IF K=32 GSB 150 スペースキーで数字を入れ替え
90 WAIT 6:GOTO 30 時間待ち:もどってループ
100 LC 1,0:FOR K=0 TO L } 数字を表示
110 ?[K]:NEXT RTN
120 [L+1]=1+RND(9) 新しい数字を設定
130 FOR K=J TO L:[K]=[K+1] 数字を左へシフト
```

20

●数字パズルゲーム 完成(2)

```
140 NEXT:RTN 数字シフトくりかえし:もどる
150 A=[X]:[X]=[X+1]:[X+1]=A 数字入れ替え
160 GSB 100 数字を表示
165 GSB 200 数字ペアをチェック
170 RTN もどる
200 P=1:FOR J=0 TO L-1 数字チェックのループ
210 IF [J]=[J+1] GSB 230:J=-1 同じ数字なら消去
220 NEXT:RTN 数字チェックここまで
230 LC J+1,0:?"**":BEEP:WAIT 10
235 GSB 120:GSB 120:GSB 100 数字消去
237 S=S+P:P=P*2 スコアを加算 数字を左へシフト
238 LC 0,3:?"S;" " スコアを表示
240 RTN もどる
```

21

手づまりまでに
何点とれるか挑戦!

まずは100点が目標
どれだけ連鎖が出せるか?

22

さらに改造しよう！

数字を変えると
けた数が変わる

```
10 CLV:CLS:VIDEO 3:L=7
```

例 L=9

```
4921757467
><
0
```

数字が10けたになる

23

さらに改造しよう！

```
210 IF ([J]+[J+1])%10=0
GSB 230:J=-1
```

となりあった数字を足して
10になると消える

数字を消すのが難しくなるが、暗算に強くなる

24